

In the oscillator example above, we could make the fixed variable R available for plotting by adding a statement

`aux myI=R`

so that `myI` is the user-accessible version of R .

3.4.2 Exercises

1. Write an ODE file for the Fibonacci recurrence

$$f_{n+1} = f_n + f_{n-1}, \quad f_0 = f_1 = 1.$$

2. Write an ODE file for the Lotka-Volterra equations

$$\frac{dx}{dt} = ax - bxy, \quad \frac{dy}{dt} = -cy + dxy$$

with positive parameters a, b, c, d . Track the quantity

$$Q = a \ln |y| + c \ln |x| - by - dx$$

with initial data $x = y = \frac{1}{2}$.

3. Write a differential equation file for the Morris-Lecar equations defined as follows:

$$C \frac{dV}{dt} = -g_L(V - E_L) - g_{Ca}m_\infty(V)(V - E_{Ca}) - g_Kw(V - E_K) + I,$$

$$\frac{dw}{dt} = \phi \frac{w_\infty(V) - w}{\tau_w(V)},$$

$$m_\infty(V) = 1/(1 + \exp(-(V - V_1)/V_2)),$$

$$w_\infty(V) = 1/(1 + \exp(-(V - V_3)/V_4)),$$

$$\tau_w(V) = 1/\cosh((V - V_3)/(2V_4)),$$

where $\phi = 0.8$, $g_K = 8$, $g_{Ca} = 4.4$, $g_L = 2$, $C = 1$, $E_K = -84$, $E_{Ca} = 120$, $E_L = -60$, $V_1 = -1.2$, $V_2 = 9$, $V_3 = 2$, $V_4 = 15$, and $I = 90$. Integrate them and analyze them along the lines that you did in Chapter 2 for the Fitzhugh-Nagumo equations. Look in the (V, w) phase-plane and compute a coexistent periodic solution and a stable fixed point. (If the ODE file is too hard for you to figure out, then, as a last resort, download the ODE file `m1ex.ode`.)

4. Write the third order system

$$x''' + ax'' + bx' + cx = x^2$$

as a system of first order equations. (**Hint:** let $y_1 = x$, $y_2 = x'$, and $y_3 = x''$. Try to write an equation for $y_3' = x'''$.) Write an ODE file for this with parameters $a = 1$,

$b = 2$, $c = 3.7$, and initial data $x = 1$, $x' = 0$, and $x'' = 0$. You will need to integrate this for a long time to see the chaotic orbit. Set the `maxstor` option to something like 20,000 (by adding the line `@ maxstor=20000` somewhere in your ODE file) and set the total time to 400. You may want to look at $x(t)$ versus $x'(t)$ (y_1 versus y_2).

5. Write an ODE file for the Rossler attractor,

$$x' = -y - z,$$

$$y' = x + ay,$$

$$z' = bx - cz + xz,$$

with parameters $a = .36$, $b = 0.4$, and $c = 4.5$ and initial data $x = 0$, $y = -4.3$, and $z = 0$. Set the total integration time to 200. Make a three-dimensional plot by clicking on the little boxes next to the three variables in the **Initial Data Window**, and then click on the `xvsv` button in the **Initial Data Window**.

6. Write and simulate the following differential equation which has three limit cycles:

$$x' = xf(r) - y,$$

$$y' = yf(r) + x,$$

$$f(r) = (1 - r)(2 - r)(3 - r),$$

$$r = \sqrt{x^2 + y^2}.$$

Set the viewing window to $[-4, 4] \times [-4, 4]$. Can you create a differential equation with five limit cycles?

7. As a bonus problem, run some of the examples in the text and some that you have written yourself to see what they do.

3.5 Discontinuous differential equations

3.5.1 Integrate-and-fire models

Consider the integrate-and-fire model for a neuron,

$$\frac{dV}{dt} = I - V,$$

with the "reset" condition that each time V hits some value V_T it is reset to 0. This leads to an oscillation when $I > V_T$. Often these models are coupled to each other through "alpha" functions, $\alpha(t) = b^2 t \exp(-bt)$. Thus, the pair satisfies

$$\frac{dV_1}{dt} = I - V_1 + g_{S_2}(t), \quad \frac{dV_2}{dt} = I - V_2 + g_{S_1}(t),$$

where, each time t^* , V_j crosses V_T , it is reset to 0 and the quantity $\alpha(t - t^*)$ is added to s_j . *XPPAUT* has a mechanism called **global flags** for incorporating such discontinuities. These are quantities that the integrator checks for zero crossings. If a zero crossing occurs, then the variables can be updated and thus discontinuously changed. The integrators are restarted so that even the adaptive step solvers will work fine. Here is the ODE file for a single integrate-and-fire oscillator, `iaandf1.ode`:

```
# integrate and fire model
v'=-v + I
global I v-vt {v=0}
par I=1.2,vt=1
@ dt=.01,ylo=0
done
```

The new line is `global I v-vt {v=0}`. This asserts that there is a global flag to check. The general syntax for flags is

```
global sign condition {event1;event2;...}
```

The condition is evaluated at each time step. If the sign is 1 and the condition changes sign from negative to positive, or if the sign is -1 and the condition changes sign from positive to negative, then each of the events is done. If the sign is 0, then the event occurs only if the condition is identically zero. Events always have the form $x = \text{expr}$, where x is one of the *variables* and expr is some expression. Thus, in the above ODE file, if $V - V_T$ changes from negative to positive (i.e., V crosses threshold from below) then V is reset to 0. The line `@ dt=.01,ylo=0` sets the time step to be 0.01 and the minimum value of the y -axis to be zero. Run this file and integrate the equations, and you will see a nice oscillation. Move the mouse inside the graphics window and hold the left button down while moving the mouse. At the bottom of the main window, the coordinates will be shown. You can use this to compute the period of the oscillation, which should be about 1.8.

Let's write an ODE file for the coupled system. The question is how to implement the alpha function. Since $b^2 t \exp(-bt)$ is a solution to

$$s'' + 2bs' + b^2s = 0, \quad s(0) = 0, \quad s'(0) = b^2,$$

this suggests letting s_j evolve according to this equation with the condition that each time the voltage crosses threshold, $s_j'(t)$ is incremented by b^2 . Here is the corresponding ODE file, `iaandf2.ode`:

```
# two integrate and fire models coupled with alpha functions
# b^2*t exp(-b*t)
# we solve these by solving a 2d ode
v1' = -v1 + i1 + g*s2
v2' = -v2 + i2 + g*s1
s1' = s1p
s1p' = -2*b*s1p - b*b*s1
s2' = s2p
s2p' = -2*b*s2p - b*b*s2
```

3.5. Discontinuous differential equations

```
#
init v1=.1
par i1=1.1,i2=1.1,vt=1,g=.2
par b=5
global I v1-vt {v1=0;s1p=s1p+b*b}
global I v2-vt {v2=0;s2p=s2p+b*b}
@ dt=.01,total=100,transient=80
@ xlo=80,xhi=100,ylo=0,nplot=2,yp2=v2
done
```

Here are some comments on the ODE file:

1. We rewrite the second order equations for s as a pair of first order ODEs.
2. We initialize $V_1 = .1$ so that it is not identical to V_2 to break the symmetry.
3. Each time V_1 crosses V_T from below, V_1 is reset to 0 and s_1' ($s1p$) is incremented by b^2 . A similar condition is set for V_2 .
4. I have set a number of options so that the user doesn't have to worry about them and can just run the integration. I have set the integration time step to 0.01, the total integration to 100 time units, and I will start storing data only after a transient of 80 time units. I set the low and high limits of the x -axis to 80 and 100, respectively. I set the lower limit of the y -axis to 0. I tell *XPPAUT* to plot two curves in the window and tell it that the second curve to plot is V_2 . (Note that, by default, the x component of a plot is time t , and the y component is the first variable defined in the file.)

Run this equation in *XPPAUT* and note that both V_1 and V_2 are synchronized. Change the initial conditions however you want (but make sure that $V_j(0) < 1$) and the solution will always go to synchrony. Change b to 1 and integrate again. The oscillations alternate; synchrony is not stable. Now change b back to 5 and change g to -2. Integrate again. Notice that synchrony is unstable. Now, change b to 1 again. Set all variables to 0 except V_1 . Set $V_1 = .1$ and integrate. You should see synchronous oscillations. Set $V_1 = .4$ and integrate; the oscillations alternate. This result was proven by van Vreeswijk, Abbott, and Ermentrout [39]. (See Figure 3.2.)

3.5.2 Clocks: Regular and irregular

There are many other examples of models that involve discontinuous right-hand sides. The simple clock is an excellent example. We model a pendulum clock as follows. It consists of a decaying sinusoid that receives a kick each time it reaches its maximum on the left-hand side. Let x be the angle of the pendulum and y be the velocity. We model the decaying sinusoid as

$$\frac{dx}{dt} = -ax - by, \quad \frac{dy}{dt} = -ay + bx$$

with the kick occurring whenever $y = 0$ and $x < 0$. X receives a kick to the left with magnitude k . Thus when $y(t) = 0$ and $x(t) < 0$, then $x(t) = x(t) - k$. The ODE file is written as